Cartesian

In[2]:= `Laplacian[f[x, y], {x, y}]`

Out[2]= $f^{(0,2)}[x, y] + f^{(2,0)}[x, y]$

In[3]:= `eq1 = x''[t] == k^2 x[t]`

Out[3]= $x''[t] == k^2\, x[t]$

In[ ]:= `DSolve[eq1, x[t], t]`

Out[ ]= $\{\{x[t] \to e^{k\,t}\, c_1 + e^{-k\,t}\, c_2\}\}$

In[8]:= `DSolve[eq1, x[t], t] // ExpToTrig`

Out[8]= $\{\{x[t] \to c_1 \text{Cosh}[k\,t] + c_2 \text{Cosh}[k\,t] + c_1 \text{Sinh}[k\,t] - c_2 \text{Sinh}[k\,t]\}\}$

In[5]:= `eq2 = x''[t] == -k^2 x[t]`

Out[5]= $x''[t] == -k^2\, x[t]$

In[6]:= `DSolve[eq2, x[t], t]`

Out[6]= $\{\{x[t] \to c_1 \text{Cos}[k\,t] + c_2 \text{Sin}[k\,t]\}\}$

In[7]:= `DSolve[eq2, x[t], t] // TrigToExp`

Out[7]= $\left\{\left\{x[t] \to \frac{1}{2} e^{-i\,k\,t}\, c_1 + \frac{1}{2} e^{i\,k\,t}\, c_1 + \frac{1}{2} i\, e^{-i\,k\,t}\, c_2 - \frac{1}{2} i\, e^{i\,k\,t}\, c_2\right\}\right\}$

Spherical

Extract pieces

In[ ]:= `Laplacian[R[r] × Y[θ, ϕ], {r, θ, ϕ}, "Spherical"]`

Out[ ]= $Y[\theta, \phi]\, R''[r] + \frac{\text{Csc}[\theta]\left(\text{Sin}[\theta]\, Y[\theta, \phi]\, R'[r] + \frac{\text{Csc}[\theta]\, R[r]\, Y^{(0,2)}[\theta,\phi]}{r} + \frac{\text{Cos}[\theta]\, R[r]\, Y^{(1,0)}[\theta,\phi]}{r}\right)}{r} + \frac{Y[\theta, \phi]\, R'[r] + \frac{R[r]\, Y^{(2,0)}[\theta,\phi]}{r}}{r}$

In[21]:= $\text{tmp1} = \frac{\text{Laplacian}[R[r] \times T[\theta] \times F[\phi], \{r, \theta, \phi\}, \text{"Spherical"}]}{R[r] \times T[\theta] \times F[\phi]}$ `// Expand`

Out[21]= $\frac{2\, R'[r]}{r\, R[r]} + \frac{\text{Cot}[\theta]\, T'[\theta]}{r^2\, T[\theta]} + \frac{\text{Csc}[\theta]^2\, F''[\phi]}{r^2\, F[\phi]} + \frac{R''[r]}{R[r]} + \frac{T''[\theta]}{r^2\, T[\theta]}$

In[25]:= `tmp2 = tmp1 (r^2 Sin[θ]^2) // Expand`

Out[25]= $\frac{2\, r\, \text{Sin}[\theta]^2\, R'[r]}{R[r]} + \frac{\text{Cos}[\theta]\, \text{Sin}[\theta]\, T'[\theta]}{T[\theta]} + \frac{F''[\phi]}{F[\phi]} + \frac{r^2\, \text{Sin}[\theta]^2\, R''[r]}{R[r]} + \frac{\text{Sin}[\theta]^2\, T''[\theta]}{T[\theta]}$

In[35]:= $\text{tmp2} = (r^2)\ \text{tmp1}\ /.\ \left\{\frac{F''[\phi]}{F[\phi]} \to -m^2\right\}$ // Expand

Out[35]= $-m^2\ \text{Csc}[\theta]^2 + \frac{2\ r\ R'[r]}{R[r]} + \frac{\text{Cot}[\theta]\ T'[\theta]}{T[\theta]} + \frac{r^2\ R''[r]}{R[r]} + \frac{T''[\theta]}{T[\theta]}$

In[45]:= `eq2 = Select[tmp2, FreeQ[#, r] &] + el (el + 1)`

Out[45]= $\text{el}\ (1 + \text{el}) - m^2\ \text{Csc}[\theta]^2 + \frac{\text{Cot}[\theta]\ T'[\theta]}{T[\theta]} + \frac{T''[\theta]}{T[\theta]}$

In[46]:= `eq3 = eq2 T[θ] // Expand`

Out[46]= $\text{el}\ T[\theta] + \text{el}^2\ T[\theta] - m^2\ \text{Csc}[\theta]^2\ T[\theta] + \text{Cot}[\theta]\ T'[\theta] + T''[\theta]$

In[47]:= `DSolve[eq3 == 0, T[θ], θ] // Simplify`

Out[47]= $\{\{T[\theta] \to c_1\ \text{LegendreP}[\text{el}, m, \text{Cos}[\theta]] + c_2\ \text{LegendreQ}[\text{el}, m, \text{Cos}[\theta]]\}\}$

In[59]:= `eq4 = Select[tmp2, FreeQ[#, θ] &] - el (el + 1)`

Out[59]= $-\text{el}\ (1 + \text{el}) + \frac{2\ r\ R'[r]}{R[r]} + \frac{r^2\ R''[r]}{R[r]}$

In[60]:= `eq5 = eq4 (R[r]) // Expand`

Out[60]= $-\text{el}\ R[r] - \text{el}^2\ R[r] + 2\ r\ R'[r] + r^2\ R''[r]$

In[68]:= `DSolve[eq5 == 0, R[r], r] // FullSimplify`

Out[68]= $\left\{\left\{R[r] \to r^{-\frac{1}{2}-\frac{1}{2} i \sqrt{\text{el}}\ \sqrt{1+\text{el}}\ \sqrt{-4-\frac{1}{\text{el}+\text{el}^2}}} \left(c_1 + r^{i \sqrt{\text{el}}\ \sqrt{1+\text{el}}\ \sqrt{-4-\frac{1}{\text{el}+\text{el}^2}}}\ c_2\right)\right\}\right\}$

In[102]:= $\text{exp1} = -\frac{1}{2}\ i\ \sqrt{\text{el}}\ \sqrt{1+\text{el}}\ \sqrt{-4-\frac{1}{\text{el}+\text{el}^2}};$

$\text{term1} = \frac{-1}{2} - \sqrt{\text{exp1}^2}$ // Expand // FullSimplify // PowerExpand // FullSimplify

Out[103]= $-1 - \text{el}$

In[92]:= $\text{exp2} = i\ \sqrt{\text{el}}\ \sqrt{1+\text{el}}\ \sqrt{-4-\frac{1}{\text{el}+\text{el}^2}};$

$\text{term2} = \sqrt{\text{exp2}^2}$ // Expand // FullSimplify // PowerExpand

Out[93]= $1 + 2\ \text{el}$

Check DEQ

In[ ]:=
```
? LegendreP
```

Out[ ]=

Symbol

LegendreP [*n*, *x*] gives the Legendre polynomial $P_n(x)$.

LegendreP [*n*, *m*, *x*] gives the associated Legendre polynomial $P_n^m(x)$.

In[105]:=
```
Clear["Global`*"]
```

$(1 - x^2)(d^2y / dx^2) - 2x(dy/dx) + n(n+1)y = 0$

In[107]:=
```
DSolve[(1 - x^2) y''[x] - 2 x y'[x] + n (n + 1) y[x] == 0, y[x], x]
```

Out[107]= {{y[x] → c_1 LegendreP[n, x] + c_2 LegendreQ[n, x]}}

In[113]:=
```
Plot[Table[LegendreQ[n, x], {n, 1, 5}] // Evaluate, {x, 0, 1}]
```

Out[113]=



In[112]:=
```
Plot[Table[LegendreP[n, x], {n, 1, 5}] // Evaluate, {x, 0, 2}]
```

Out[112]=



In[114]:=
```
tmp = LegendreP[n, z]
```

Out[114]= LegendreP[n, z]

In[115]:= `(1 - z^2) D[tmp, {z, 2}] - 2 z D[tmp, z] + n (n + 1) tmp // Simplify`

Out[115]= 0

In[116]:= `t1 = (1 - z^2) D[tmp, {z, 2}]`

Out[116]= $$\left(1-z^2\right)\left(\frac{\left(-n-n^2+2 z^2+3 n z^2+n^2 z^2\right) \text{LegendreP}[n, z]}{\left(-1+z^2\right)^2}-\frac{2 (1+n) z\, \text{LegendreP}[1+n, z]}{\left(-1+z^2\right)^2}\right)$$

In[117]:= `t2 = - 2 z D[tmp, z]`

Out[117]= $$-\frac{2 z ((-1-n) z\, \text{LegendreP}[n, z]+(1+n) \text{LegendreP}[1+n, z])}{-1+z^2}$$

In[118]:= `t3 = n (n + 1) tmp`

Out[118]= n (1 + n) LegendreP[n, z]

In[119]:= `t1 + t2 + t3 // Simplify`

Out[119]= 0

Cylindrical

Extract pieces

In[ ]:= `Laplacian[f[r, θ, z], {r, θ, z}, "Cylindrical"]`

Out[ ]= $$f^{(0,0,2)}[r, \theta, z]+\frac{\frac{f^{(0,2,0)}[r,\theta,z]}{r}+f^{(1,0,0)}[r, \theta, z]}{r}+f^{(2,0,0)}[r, \theta, z]$$

In[127]:= $$\text{tmp1} = \frac{\text{Laplacian}[R[r] \times T[\theta] \times Z[z], \{r, \theta, z\}, \text{"Cylindrical"}]}{R[r] \times T[\theta] \times Z[z]} \text{ // Expand}$$

Out[127]= $$\frac{R'[r]}{r\, R[r]}+\frac{R''[r]}{R[r]}+\frac{T''[\theta]}{r^2\, T[\theta]}+\frac{Z''[z]}{Z[z]}$$

In[125]:= $$\text{zeq} = \frac{Z''[z]}{Z[z]} == -k^2;$$

`DSolve[zeq, Z[z], z]`

Out[126]= $\{\{Z[z] \to c_1 \text{Cos}[k z] + c_2 \text{Sin}[k z]\}\}$

In[130]:= $$\text{tmp2} = r^2\ \text{tmp1} \ /. \left\{\frac{Z''[z]}{Z[z]} \to -k^2\right\} \text{ // Expand}$$

Out[130]= $$-k^2 r^2+\frac{r\, R'[r]}{R[r]}+\frac{r^2\, R''[r]}{R[r]}+\frac{T''[\theta]}{T[\theta]}$$

In[158]:=
```
teq = Select[tmp2, FreeQ[#, R[r]] &]
```
Out[158]= $-k^2 r^2 + \frac{T''[\theta]}{T[\theta]}$

In[160]:=
```
teq = Select[tmp2, FreeQ[#, R[r]] &] + (n^2 + k^2 r^2)
```
Out[160]= $n^2 + \frac{T''[\theta]}{T[\theta]}$

In[163]:=
```
teq2 = T[θ] teq // Expand
```
Out[163]= $n^2\, T[\theta] + T''[\theta]$

In[164]:=
```
DSolve[teq2 == 0, T[θ], θ]
```
Out[164]= $\{\{T[\theta] \to c_1 \operatorname{Cos}[n\,\theta] + c_2 \operatorname{Sin}[n\,\theta]\}\}$

In[165]:=
```
req = Select[tmp2, FreeQ[#, T[θ]] &]
```
Out[165]= $-k^2 r^2 + \frac{r\,R'[r]}{R[r]} + \frac{r^2\,R''[r]}{R[r]}$

In[178]:=
```
req = Select[tmp2, FreeQ[#, T[θ]] &] - n^2
```
Out[178]= $-n^2 - k^2 r^2 + \frac{r\,R'[r]}{R[r]} + \frac{r^2\,R''[r]}{R[r]}$

In[179]:=
```
req2  =  req R[r] // Expand
```
Out[179]= $-n^2\, R[r] - k^2 r^2\, R[r] + r\,R'[r] + r^2\, R''[r]$

In[180]:=
```
DSolve[req2 == 0, R[r], r]
```
Out[180]= $\{\{R[r] \to \operatorname{BesselJ}[n, -i\,k\,r]\, c_1 + \operatorname{BesselY}[n, -i\,k\,r]\, c_2\}\}$

In[186]:=
```
Plot[Table[BesselJ[n, x], {n, 1, 5}] // Evaluate, {x, 0, 10}]
```
Out[186]=



In[185]:=
```
Plot[Table[BesselY[n, x], {n, 1, 5}] // Evaluate , {x, 0, 10}]
```

Out[185]=



Check DEQ

In[ ]:=
```
? BesselJ
```

Out[ ]=

Symbol

BesselJ [n, z] gives the Bessel function of the first kind $J_n(z)$.

In[ ]:=
```
Clear["Global`*"]
```

In[ ]:= $z^2 y'' + z y' + (z^2 - n^2) y = 0$

Set : Tag Plus in $y(-n^2 + z^2) + z y' + z^2 y''$ is Protected .

Out[ ]= 0

In[187]:= DSolve$\left[z^2\, y''[z] + z\, y'[z] + (z^2 - n^2)\, y[z] == 0,\ y[z],\ z\right]$

Out[187]= $\{\{y[z] \to \text{BesselJ}[n, z]\, c_1 + \text{BesselY}[n, z]\, c_2\}\}$

In[188]:=
```
tmp = BesselJ[n, z]
```

Out[188]= BesselJ[n, z]

In[189]:= z^2 D[tmp, {z, 2}] + z D[tmp, z] + $(z^2 - n^2)$ tmp // FullSimplify

Out[189]= 0

In[190]:= t1 = z^2 D[tmp, {z, 2}]

Out[190]= $z^2 \left(-\frac{\text{BesselJ}[-1+n, z]}{z} + \frac{(n + n^2 - z^2)\,\text{BesselJ}[n, z]}{z^2}\right)$

In[191]:=
```
t2 = z D[tmp, z]
```

Out[191]= $\frac{1}{2} z\,(\text{BesselJ}[-1+n, z] - \text{BesselJ}[1+n, z])$

In[192]:=
```
t3 = (z^2 - n^2) tmp
```

Out[192]=
$(-n^2 + z^2)$ BesselJ[n, z]

In[193]:=
```
t1 + t2 + t3 // FullSimplify
```

Out[193]=
0